

Permission verification vulnerability: Analysis of the SushiSwap Attack



Apr 11, 2023

SharkTeam, a leading blockchain security service team, offers smart contract audit services for developers. To satisfy the demands of different clients, the smart contract audit services provide both manual auditing and automated auditing.

We implement almost 200 auditing contents that cover four aspects: high-level language layer, virtual machine layer, blockchain layer, and business logic layer, ensuring that smart contracts are completely guaranteed and Safe.

(2) In the processRoute function, call processMyERC20 through the input parameter route.

```
function processRouteInternal(
    address tokenIn,
    uint256 amountIn,
    address tokenOut,
    uint256 amountOutMin,
    address to,
    bytes memory route
) private returns (uint256 amountOut) {
    uint256 balanceInInitial = tokenIn == NATIVE_ADDRESS ? address(this).balance : IERC20(tokenIn).balanceOf(msg.sender);
    uint256 balanceOutInitial = tokenOut == NATIVE_ADDRESS ? address(to).balance : IERC20(tokenOut).balanceOf(to);

    uint256 stream = Inputstream.createStream(route);
    while (stream.isNotEmpty()) {
        uint8 commandCode = stream.readUint8();
        if (commandCode == 1) processMyERC20(stream);
        else if (commandCode == 2) processUserERC20(stream, amountIn);
        else if (commandCode == 3) processNative(stream);
        else if (commandCode == 4) processOnePool(stream);
        else if (commandCode == 5) processInsideBento(stream);
        else revert('RouteProcessor: Unknown command code');
    }
}

function processRoute(
    address tokenIn,
    uint256 amountIn,
    address tokenOut,
    uint256 amountOutMin,
    address to,
    bytes memory route
) external payable lock returns (uint256 amountOut) {
    return processRouteInternal(tokenIn, amountIn, tokenOut, amountOutMin, to, route);
}
```

(3) The stream parameter controlled by the attacker is called through a series of function calls such as processMyERC20 -> distributeAndSwap -> swap, and finally calls the swapUniV3 function.

```
function distributeAndSwap(uint256 stream, address from, address tokenIn, uint256 amountTotal) private {
    uint8 num = stream.readUint8();
    unchecked {
        for (uint256 i = 0; i < num; ++i) {
            uint16 share = stream.readUint16();
            uint256 amount = (amountTotal * share) / 65535;
            amountTotal -= amount;
            swap(stream, from, tokenIn, amount);
        }
    }
}

function swap(uint256 stream, address from, address tokenIn, uint256 amountIn) private {
    uint8 poolType = stream.readUint8();
    if (poolType == 0) swapUniV2(stream, from, tokenIn, amountIn);
    else if (poolType == 1) swapUniV3(stream, from, tokenIn, amountIn);
    else if (poolType == 2) wrapNative(stream, from, tokenIn, amountIn);
    else if (poolType == 3) bentoBridge(stream, from, tokenIn, amountIn);
    else if (poolType == 4) swapTrident(stream, from, tokenIn, amountIn);
    else if (poolType == 5) swapTridentCL(stream, from, tokenIn, amountIn);
    else revert('RouteProcessor: Unknown pool type');
}

function processMyERC20(uint256 stream) private {
    address token = stream.readAddress();
    uint256 amountTotal = IERC20(token).balanceOf(address(this));
    unchecked {
        if (amountTotal > 0) amountTotal -= 1; // slot undrain protection
    }
    distributeAndSwap(stream, address(this), token, amountTotal);
}
```

(4) In the swapUniV3 function, the pool value is parsed from the stream parameter value controlled by the attacker, which is actually the attack contract. The parsed pool value is assigned to lastCalledPool. At this time, lastCalledPool is the malicious value controlled by the attacker (that is, the attack contract), and then the swap function defined by the pool, which is the attack contract, is called.

```
function swapUniV3(uint256 stream, address from, address tokenIn, uint256 amountIn) private {
    address pool = stream.readAddress();
    bool zeroForOne = stream.readUint8() > 0;
    address recipient = stream.readAddress();

    lastCalledPool = pool;
    IUniswapV3Pool(pool).swap(
        recipient,
        zeroForOne,
        int256(amountIn),
        zeroForOne ? MIN_SQRT_RATIO + 1 : MAX_SQRT_RATIO - 1,
        abi.encode(tokenIn, from)
    );
    require(lastCalledPool == IMPOSSIBLE_POOL_ADDRESS, 'RouteProcessor.swapUniV3: unexpected'); // Just to be sure
}
```

(5) The `uniswapV3SwapCallback` function in the vulnerable contract was called back through the `swap` function, and 100 ETHs were successfully extracted.

[illegible]

Vulnerability analysis:

In the `swapUniV3` function, the `pool` value comes from the `stream` parameter value maliciously constructed by the attacker, which is actually the attack contract, and then calls the custom swap function in the attack contract.

```
function swapUniV3(uint256 stream, address from, address tokenIn, uint256 amountIn) private {
    address pool = stream.readAddress();
    bool zeroForOne = stream.readUint8() > 0;
    address recipient = stream.readAddress();

    lastCalledPool = pool;
    IUniswapV3Pool(pool).swap(
        recipient,
        zeroForOne,
        int256(amountIn),
        zeroForOne ? MIN_SQRT_RATIO + 1 : MAX_SQRT_RATIO - 1,
        abi.encode(tokenIn, from)
    );
    require(lastCalledPool == IMPOSSIBLE_POOL_ADDRESS, 'RouteProcessor.swapUniV3: unexpected'); // Just to be sure
}
```

In the swap function, the uniswapV3SwapCallback function in the vulnerable contract is called back.

```
function uniswapV3SwapCallback(
    int256 amount0Delta,
    int256 amount1Delta,
    bytes calldata data
) external {
    require(msg.sender == lastCalledPool, 'RouteProcessor.uniswapV3SwapCallback: call from unknown source');
    lastCalledPool = IMPOSSIBLE_POOL_ADDRESS;
    (address tokenIn, address from) = abi.decode(data, (address, address));
    int256 amount = amount0Delta > 0 ? amount0Delta : amount1Delta;
    require(amount > 0, 'RouteProcessor.uniswapV3SwapCallback: not positive amount');

    if (from == address(this)) IERC20(tokenIn).safeTransfer(msg.sender, uint256(amount));
    else IERC20(tokenIn).safeTransferFrom(from, msg.sender, uint256(amount));
}
```

Since lastCalledPool is set to the pool value in the swapUniV3 function (that is, the address of the attack contract), when the uniswapV3SwapCallback function is called back, msg.sender == lastCalledPool is true, directly bypassing the authority check on whether msg.sender is a real pool address, and passed the parameters through the parameter construction token, stealing the WETH authorized by the account to the vulnerable contract that has not been used. Then set lastCalledPool to the constant IMPOSSIBLE_POOL_ADDRESS, thus passing the last require check of lastCalledPool==IMPOSSIBLE_POOL_ADDRESS in the swapUniV3 function.

Vulnerability Summary:

The root cause of this incident is that the vulnerable contract (RouteProcessor2) did not check whether the callback caller was a V3 pool deployed by the uniswapV3 factory, but only checked whether it was the original lastCalledPool value, resulting in the theft of too much unused funds authorized to the RouteProcessor2 contract for the account.

2. Security Recommendations

In response to this attack, we should follow the following precautions during the development process:

- (1) In each business module where the user interacts with the contract, the principle of least authority should be followed, and the actual amount of a single transaction should be reasonably authorized to avoid the loss of account funds caused by excessive amount authorization.
- (2) In a callback function similar to `uniswapV3SwapCallback`, it should be verified whether the caller is the V3 pool deployed by `uniswapV3` factory.

About Us

Our vision is to improve security globally. We believe that by building this security barrier, we can significantly improve lives around the world. SharkTeam composes of members with many years of cyber security experiences and blockchain, team members are based in Suzhou, Beijing, Nanjing and Silicon Valley, proficient in the underlying theories of blockchain and smart contracts, and we provide comprehensive services including threat modeling, smart contract auditing, emergency response, etc. SharkTeam has established strategic and long-term cooperations with key players in many areas of the blockchain ecosystem, such as Huobi Global, OKX, polygon, Polkadot, imToken, ChainIDE, etc

Twitter: <https://twitter.com/sharkteamorg>

Discord: <https://discord.gg/jGH9xXCjDZ>

Telegram: <https://t.me/sharkteamorg>

web: <https://www.sharkteam.org/>



SharkTeam

In Math, We Trust!



<https://sharkteam.org>



<https://t.me/sharkteamorg>



<https://twitter.com/sharkteamorg>